

Creating Word Tables using PROC REPORT and ODS RTF

Carey G. Smoak, Roche Molecular Systems, Inc., Pleasanton, CA

ABSTRACT

With the introduction of the ODS RTF destination, programmers now have the ability to create Word tables using SAS®. This paper focuses specifically on the use of the ODS RTF destination with PROC REPORT to create Word tables. Demographic data from a research study is used to illustrate how to create Word tables. Some basic methods for generating Word tables are demonstrated. Several problems are illustrated and solutions are presented. Version 8.2 of SAS on a Windows platform was used.

INTRODUCTION

The RTF destination (starting with version 7 of SAS) for ODS allows the user to create Word tables by using SAS. While ODS RTF is a wonderful tool, it can also be frustrating for the user to learn its capabilities and limitations. In this paper, I present some basic methods to help the user produce Word tables using ODS RTF and PROC REPORT. This paper is not intended to be a comprehensive explanation of how to use ODS RTF with PROC REPORT. Rather, it is designed to help the user with certain tasks common in the biotechnology/diagnostic/pharmaceutical industries and how to overcome certain problems with ODS RTF with PROC REPORT. Some of the common tasks that I demonstrate in this paper are

- Creating headers, titles and footnotes
- Page numbering (Page X of Y)
- The bodytitle option and its effect on headers, titles, footnotes and page numbering
- Breaking a page

The need to produce tables that can be inserted directly into document text (in-text tables) for study reports, etc., motivated me to learn how to use PROC REPORT and ODS RTF to produce Word tables. While the methods presented here can be used to produce tables and listings for study reports and other documents, I find this method most useful for in-text tables, which typically have few columns. For example, ODS RTF makes it difficult to pinpoint exactly where you want information on the page (ODS RTF does not understand code like "line @15 ..." in a compute block). Therefore, I find PROC REPORT and ODS RTF most useful for producing tables with only a few columns. (Often, tables and listings for a final report have many columns. In this case, using PROC REPORT or data _null_ to create a text file and then inserting the text file into Word may be easier.)

I learned most of the information presented in this paper from two SAS live web classes that I took:

- Creating Detail and Summary Reports with the REPORT Procedure (a basic class on the syntax of PROC REPORT)
- Customizing Output from the REPORT Procedure with ODS (this class focused specifically on using ODS with PROC REPORT)

DESCRIPTION OF DATA

The data in this paper is demographic data from a research study. Five demographic variables were collected in this study: age (years), race, sex, height (cm) and weight (kg). There were also five treatment groups in the study.

In the example programs presented in this paper, the goal is to produce a summary table of the demographic variables for each treatment group. The number of observations, mean, standard deviation, minimum and maximum are presented for continuous variables (age, height and weight) and the number of observations and percent are presented for categorical variables (race and sex). The goal is to produce a table that has the treatment groups listed across the top of the table and the demographic factors listed down the left side of the table.

PROC REPORT

First, let us look at how to use PROC REPORT to produce a demography summary table as a text file. In the program below, a table called demog_rept.txt will be created using PROC REPORT. This code will be modified in Examples 1-3 to show how a Word table can be created with ODS RTF.

```
/** Create filename for output **/  
filename outfile "&path\&dirname\demog_rept.txt";  
  
proc printto file=outfile new;  
run;  
  
options nodate nonumber;  
  
proc report data=final nowd headline headskip ls=132 ps=49 missing;  
  
    columns factor category _name_ grp_1 grp_2 grp_3 grp_4 grp_5;
```

```

define factor / "" order order=data width=10;
define category / "" width=13;
define _name_ / "" width=11;
define grp_1 / "Group 1" width=11;
define grp_2 / "Group 2" width=11;
define grp_3 / "Group 3" width=11;
define grp_4 / "Group 4" width=11;
define grp_5 / "Group 5" width=11;

break after factor / skip;

compute before _page_;
  line @15 "Roche Molecular Systems, Inc.";
  line @15 "Study No.: 999";
  line @15 "";
  line @50 "Table 1: Demography Summary";
  line @15 "";
  line @15 103*'-' ;
endcomp;

compute after _page_;
  line @15 103*'-' ;
  line @15 "";
  line @15 "Output: &path\&dirname\demog_rept.txt";
  line @15 "Table run: &dtm";
endcomp;

run;

proc printto;
run;

```

The output for this report is not shown in this paper.

PROC REPORT & ODS RTF: EXAMPLE 1

Now let's modify this program to produce a Word table. First, we add an ODS RTF statement and name the output as demog_rprt1.rtf. Next, note that we have nodate and nonumber as options, which is typical for reports in the biotechnology/diagnostics/pharmaceutical industries. This is important, because in Example 3, I will use an option called bodytitle.

Next, some style statements are added to the PROC REPORT statement. When style statements are used at this level, they set up in the general format of the Word table. These general formats can be overridden by including style statements in the define statements. For example, you can define just=left for a header in the PROC REPORT statement and then override the header with just=center in a define statement.

Next, the columns of the report are defined by using the columns and define statements. In the columns statement, you will notice a variable called mypage. This variable tells the table when to break and go to the next page.

Finally, note the use of compute blocks to print the header, title and footnotes by using line statements. In these compute blocks, I am using RTF control strings to control whether to left-justify (/par/ql) a line or to center (/par/qc) it. Note that in the PROC REPORT statement I define just=left for style(lines). That is why I needed to use "/par/qc" to center the line for "Table 1: Demography Summary". In order to use these RTF control strings, you must use protectspecialcharacters=off in a style statement.

```

ods rtf file = "&path\&dirname\demog_rpt1.rtf";

options nodate nonumber;

proc report data=final nowd split='*' missing
  style(report)={just=center}
  style(lines)=header{background=white asis=on font_size=12pt font_face="TimesRoman"
    font_weight=bold just=left}
  style(header)=header{background=white font_size=10pt font_face="TimesRoman" frame=box
    font_weight=bold}
  style(column)=header{background=white font_size=10pt font_face="TimesRoman"
    font_weight=medium};

columns factor category _name_ grp_1 grp_2 grp_3 grp_4 grp_5;

define factor / "Demographic*Factor" order order=data width=10;

```

```

define category / "" width=13 style(column)={just=center};
define _name_ / "Statistic" width=11 style(column)={just=center};
define grp_1 / "Group 1" width=11 style(column)={just=center};
define grp_2 / "Group 2" width=11 style(column)={just=center};
define grp_3 / "Group 3" width=11 style(column)={just=center};
define grp_4 / "Group 4" width=11 style(column)={just=center};
define grp_5 / "Group 5" width=11 style(column)={just=center};

break after factor / skip;

compute before _page_ / style=[protectspecialchars=off];
  line "Roche Molecular Systems, Inc.";
  line "Study No.: 999";
  line "";
  line "\par\qc{Table 1: Demography Summary}";
  line "\par\ql{ }";
endcomp;

compute after _page_ / style=[protectspecialchars=on];
  line "";
  line "Output: &path\&dirname\demog_rpt1.rtf";
  line "Table run: &dtm";
endcomp;

run;

ods rtf close;

```

The output for this report is shown at the end of this paper (Example 1).

PROC REPORT & ODS RTF: EXAMPLE 2

The difference between Example 1 and Example 2 is that in example 2, I use title and footnote statements to define the header, title, and footnotes instead of using compute blocks with line statements. In this example I do not have to use RTF control strings. Instead I can use j=l (left justification) and j=c (centered) to control the justification. Note that the header, title, and footnotes appear on each page of the table, but they are in the header and footer of the document.

In this example, I also show how to break a page by using a variable that I created called mypage. This variable was created in the data step which created work.final:

```

data final;
  ...

  counter = _n_;
  mypage = 1;

  if counter > 17 then do;
    counter = 1;
    mypage + 1;
  end;

run;

```

What I did was to look at the output from demog_rept.txt (not shown) and count the number of lines which would be required to break after sex. So when the counter is greater than 17, I reset it to 1 and increment mypage by 1.

This example also shows how to use RTF control strings (see footnote 3 below) to print Page X of Y. I know that the code looks unsightly, but it works. When you open the RTF file, the page numbering will sometimes look backwards (Page Y of X), however, it will print out correctly as Page X of Y.

You will notice on the second page of Example 2 that there is a narrow column between "Demographic Factor" and "Statistic" and that the column for "Statistic" is not wide enough to correctly print the word "Statistic". I will show how to correct these problems in Example 3.

```

ods rtf file = "&path\&dirname\demog_rpt2.rtf";

options nodate nonumber;

proc report data=final nowd split='*' missing

```

```

style(report)={just=center}
style(lines)=header{background=white asis=on font_size=12pt font_face="TimesRoman"
font_weight=bold just=left}
style(header)=header{background=white font_size=10pt font_face="TimesRoman" frame=box
font_weight=bold}
style(column)=header{background=white font_size=10pt font_face="TimesRoman"
font_weight=medium};

columns mypage factor category _name_ grp_1 grp_2 grp_3 grp_4 grp_5;

define mypage / order noprint;
define factor / "Demographic*Factor" order order=data width=10;
define category / " " width=13 style(column)={just=center};
define _name_ / "Statistic" width=11 style(column)={just=center};
define grp_1 / "Group 1" width=11 style(column)={just=center};
define grp_2 / "Group 2" width=11 style(column)={just=center};
define grp_3 / "Group 3" width=11 style(column)={just=center};
define grp_4 / "Group 4" width=11 style(column)={just=center};
define grp_5 / "Group 5" width=11 style(column)={just=center};

break after mypage / page;

break after factor / skip;

title1 j=1 "Roche Molecular Systems, Inc.";
title2 j=1 "Study No.: 999";
title3;
title4 j=c "Table 1: Demography Summary";
title5;

footnote1;
footnote2 j=1 "Output: &path&dirname\demog_rpt2.rtf";
footnote3 j=1 "Table run: &dtm";
j=r "{Page} {\field{\*\fldinst { PAGE }}}\~{of}\~{\field {\*\fldinst
{ NUMPAGES }}}";

run;

ods rtf close;

```

The output for this report is shown at the end of this paper (Example 2).

PROC REPORT & ODS RTF: EXAMPLE 3

In this example, I now use a special option called bodytitle. In Example 2, the text from the title and footnote statements are placed in the header and footer of the Word table. The bodytitle option allows the text from the title and footnote statements to be placed in the body of the table. However, as previously mentioned, the bodytitle option does not allow you to use both nodate and nonumber. Because I use the number option, a number is placed in the header in the upper right corner of the output (see Examples 3 at the end of this paper). Notice also that the bodytitle option places the footnote only on the last page of the table. Therefore, you should not use the bodytitle option if you want the footnote to appear on every page of the table. Another problem with bodytitle is that the page numbering prints out as "Page 1 of". This is because I used the portrait orientation for the table. When landscape orientation is used the bodytitle option can be used with the RTF control strings for page numbering.

In Example 2, in the second page of the output, there is a narrow blank column between the "Demographic Factor" and "Statistic" columns and that the word "Statistic" does not print correctly in the column. Here I have increased the outputwidth to 7 inches and added cellwidth statements (.75 inch). This will correct this problem.

```

ods rtf file = "&path&dirname\demog_rpt4.rtf" bodytitle;

options nodate number;

proc report data=final nowd split='*' missing
style(report)={just=center outputwidth=7 in}
style(lines)=header{background=white asis=on font_size=12pt font_face="TimesRoman"
font_weight=bold just=left}
style(header)=header{background=white font_size=10pt font_face="TimesRoman" frame=box
font_weight=bold}
style(column)=header{background=white font_size=10pt font_face="TimesRoman"

```

```

font_weight=medium};

columns mypage factor category _name_ grp_1 grp_2 grp_3 grp_4 grp_5;

define mypage / order noprint;
define factor / "Demographic*Factor" order order=data
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define category / ""
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define _name_ / "Statistic"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define grp_1 / "Group 1"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define grp_2 / "Group 2"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define grp_3 / "Group 3"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define grp_4 / "Group 4"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};
define grp_5 / "Group 5"
style(column)={just=center cellwidth=.75 in}
style(header)={cellwidth=.75 in};

break after mypage / page;

break after factor / skip;

title1 j=1 "Roche Molecular Systems, Inc.";
title2 j=1 "Study No.: 999";
title3;
title4 j=c "Table 1: Demography Summary";
title5;

footnote1;
footnote2 j=1 "Output: &path&\&dirname\demog_rpt4.rtf";
footnote3 j=1 "Table run: &dtm";
j=r "{Page} {\field{\*\fldinst { PAGE }}}\~{of}\~{\field {\*\fldinst
{ NUMPAGES }}}";

run;

ods rtf close;

```

The output for this report is shown at the end of this paper (Example 3).

CONCLUSION

The methods presented here are intended to help you get started in creating Word tables using ODS RTF and PROC REPORT. However, there are some problems that you need to be aware of in creating tables with this tool. This paper was designed to help you get started in creating Word tables and help you to overcome some common problems. Specifically, I have shown you methods for:

- Creating headers, titles, and footnotes
- Page numbering (Page X of Y)
- Using the bodytitle option and its effect on headers, titles, footnotes, and page numbering
- Breaking a page

For general information on PROC REPORT, I would recommend the beginning tutorial by Ma, Schlotzhauer, and Ilieva (2002). For more advanced methods of ODS RTF and PROC REPORT, I would recommend the articles by Hamilton (2003), Hull (2001) and Shannon (2002). For more information on RTF specifications, see the Web site indicated in the References section.

REFERENCES

Hamilton P. (2003), "ODS to RTF: Tips and Tricks," Proceedings of the Twenty-eighth Annual SAS User Group International Conference, Cary, NC: SAS Institute, Inc., paper 24-28.

Hull B. (2001), "Now There is an Easy Way to Get to Word, Just Use PROC TEMPLATE, PROC REPORT, and ODS RTF" Proceedings of the Twenty-sixth Annual SAS User Group International Conference, Cary, NC: SAS Institute, Inc., paper 163-26.

Ma J.M., Schlotzhauer S., Ilieva M. (2002), "Quick Results in PROC REPORT" Proceedings of the Twenty-seventh Annual SAS User Group International Conference, Cary, NC: SAS Institute, Inc., paper 59-27.

Rich –Text Format Specification v. 1.2 from www.dubois.ws/software/RTF/RTF-Spec-1.7.rtf

Shannon D. (2002), "To ODS RTF and Beyond," Proceedings of the Twenty-seventh Annual SAS User Group International Conference, Cary, NC: SAS Institute, Inc., paper 1-27.

ACKNOWLEDGMENTS

I would like to thank Cynthia Zender, SAS Institute, for her assistance in helping me learn various aspects of using ODS RTF with PROC REPORT.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Carey G. Smoak
Roche Molecular Systems, Inc.
4300 Hacienda Drive
Pleasanton, CA 94588
Tel. 925-730-8033
Fax 925-225-0195
carey.smoak@roche.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

EXAMPLE 1

Header left justified

Roche Molecular Systems, Inc. Study No.: 999							
Table 1: Demography Summary							
Demographic Factor		Statistic	Group 1	Group 2	Group 3	Group 4	Group 5
AGE		N	33	20	20	45	41
		Mean	41.8	41.9	43.1	42	41.6
		Std	5.9	4.8	6.7	6.5	5.7
		Min	31	34	32	18	27
		Max	55	52	55	54	57
RACE	BLACK	N	4	2	0	4	4
		Percent	12.1	10	0	8.9	9.8
	CAUCASIAN	N	26	18	19	40	36
		Percent	78.8	90	95	88.9	87.8
	HISPANIC	N	2	0	0	1	1
		Percent	6.1	0	0	2.2	2.4
	ORIENTAL	N	1	0	1	0	0
		Percent	3	0	5	0	0
SEX	FEMALE	N	7	7	6	8	6
		Percent	21.2	35	30	17.8	14.6
	MALE	N	26	13	14	37	35
		Percent	78.8	65	70	82.2	85.4
HEIGHT		N	32	20	20	45	41
		Mean	175.7	175.7	174.9	176.9	175
		Std	11	8.1	12.5	7.5	9
		Min	152	160	152	157	150
		Max	193	188	193	188	188
WEIGHT		N	33	20	20	45	41
		Mean	86.8	82.8	87.6	87.5	85.9
		Std	15.5	20.5	18.1	16.6	20.2
		Min	54	46	54	54	40
		Max	116	123	123	128	144
Output: u:\sas\basas\demog_rpt1_por.rtf Table run: 16OCT2003 11:11							

Title centered

Footnote left justified

EXAMPLE 2

Roche Molecular Systems, Inc.
Study No.: 999

Header in Word
"header"

Table 1: Demography Summary

Title in Word
"header"

Demographic Factor		Statistic	Group 1	Group 2	Group 3	Group 4	Group 5
AGE		N	33	20	20	45	41
		Mean	41.8	41.9	43.1	42	41.6
		Std	5.9	4.8	6.7	6.5	5.7
		Min	31	34	32	18	27
		Max	55	52	55	54	57
RACE	BLACK	N	4	2	0	4	4
		Percent	12.1	10	0	8.9	9.8
	CAUCASIAN	N	26	18	19	40	36
		Percent	78.8	90	95	88.9	87.8
	HISPANIC	N	2	0	0	1	1
		Percent	6.1	0	0	2.2	2.4
ORIENTAL	N	1	0	1	0	0	
	Percent	3	0	5	0	0	
SEX	FEMALE	N	7	7	6	8	6
		Percent	21.2	35	30	17.8	14.6
	MALE	N	26	13	14	37	35
		Percent	78.8	65	70	82.2	85.4

Footnote in
Word "footer"

Output: u:\sas\basas\demog_rpt2_por.rtf
Table run: 16OCT2003 11:11

Page X of Y

Page 1 of 2

EXAMPLE 2 (continued)

Roche Molecular Systems, Inc.
Study No.: 999

Column not wide enough

Narrow column

Table Demography Summary

Demographic Factor	Statistic	Group 1	Group 2	Group 3	Group 4	Group 5
HEIGHT	N	32	20	20	45	41
	Mean	175.7	175.7	174.9	176.9	175
	Std	11	8.1	12.5	7.5	9
	Min	152	160	152	157	150
	Max	193	188	193	188	188
WEIGHT	N	33	20	20	45	41
	Mean	86.8	82.8	87.6	87.5	85.9
	Std	15.5	20.5	18.1	16.6	20.2
	Min	54	46	54	54	40
	Max	116	123	123	128	144

EXAMPLE 3

Roche Molecular Systems, Inc.
Study No.: 999

Header with
"body" of table

1
Number
in Word
"header"

Title with
"body" of table

Table 1: Demography Summary

Demographic Factor		Statistic	Group 1	Group 2	Group 3	Group 4	Group 5
AGE		N	33	20	20	45	41
		Mean	41.8	41.9	43.1	42	41.6
		Std	5.9	4.8	6.7	6.5	5.7
		Min	31	34	32	18	27
		Max	55	52	55	54	57
RACE	BLACK	N	4	2	0	4	4
		Percent	12.1	10	0	8.9	9.8
	CAUCASIAN	N	26	18	19	40	36
		Percent	78.8	90	95	88.9	87.8
	HISPANIC	N	2	0	0	1	1
		Percent	6.1	0	0	2.2	2.4
	ORIENTAL	N	1	0	1	0	0
		Percent	3	0	5	0	0
SEX	FEMALE	N	7	7	6	8	6
		Percent	21.2	35	30	17.8	14.6
	MALE	N	26	13	14	37	35
		Percent	78.8	65	70	82.2	85.4

No footnote in
Word "footer"

EXAMPLE 3 (continued)

Roche Molecular Systems, Inc.
Study No.: 999

Table 1: Demography Summary

Demographic Factor		Statistic	Group 1	Group 2	Group 3	Group 4	Group 5
HEIGHT		N	32	20	20	45	41
		Mean	175.7	175.7	174.9	176.9	175
		Std	11	8.1	12.5	7.5	9
		Min	152	160	152	157	150
		Max	193	188	193	188	188
WEIGHT		N	33	20	20	45	41
		Mean	86.8	82.8	87.6	87.5	85.9
		Std	15.5	20.5	18.1	16.6	20.2
		Min	54	46	54	54	40
		Max	116	123	123	128	144

Column is wider

Footnote at end of table and with "body" of table

Page X of _

Output: u:\sas\basas\demog_rpt3_por.rtf
Table run: 16OCT2003 11:11

Page 2 of 2